Lezione 32 (ore 54,55) - 18 maggio 2016, 16:30 - 18:30 B23

(32.01) Osservazione.

La convergenza del metodo di Eulero si ottiene considerando che:

$$E \to 0 \quad \Rightarrow \quad \frac{1}{2}\ C'' \sqrt{\frac{2E}{\lambda}} \quad \frac{e^{L\ (tf\ -\ t0)} - 1}{L} \to 0$$

Per E → 0 è ragionevole aspettarsi, più precisamente, che:

- N *tende a infinito come* $1 / \sqrt{E}$;

  (Infatti:

  $$\mathrm{int}(\ (tf - t0) \sqrt{\frac{\lambda}{2E}}\ ) = Nmin \leqslant N \leqslant Nmax$$

  e sia Nmin che Nmax tendono a infinito come $1 / \sqrt{E}$.)

- *Per ogni* k: ET(k) *tende a zero come* $\sqrt{E}$.

(32.02) Esempio.

(1) Consideriamo l'equazione differenziale:

$$x'(t) = sen(x(t))$$

In questo caso F(t,x) = sen(x) e le condizioni (a) e (b) sono soddisfatte:

- Tutte le derivate parziali prime di F(t,x) esistono e sono funzioni continue di t ed x;
- La funzione F(t,x) è lipschitziana di costante L = 1 rispetto ad x;

  (Infatti: Per ogni x',x" in R esiste z tra x' ed x" tale che (Teorema di Lagrange):

  $$|\ F(t,x') - F(t,x")\ | = |\ sen(x') - sen(x")\ | = |\ cos(z)\ |\ |x' - x"|$$

  da cui: $|\ F(t,x') - F(t,x")\ | \leqslant |x' - x"|$.)

- La funzione F(t,0) = sen(0) è limitata su R;
- Si ha:

  $$G(t,x) = cos(x)\ sen(x) \quad \Rightarrow \quad \text{per ogni } (t,x):\ ||\ G(t,x)\ || \leqslant 1$$

Il Teorema di convergenza (30.01) *è utilizzabile*.

(2) Consideriamo l'equazione differenziale:

x'(t) = x(t)

In questo caso F(t,x) = x, la condizione (a) è soddisfatta:

- Tutte le derivate parziali prime di F(t,x) esistono e sono funzioni continue di t ed x;
- La funzione F(t,x) è lipschitziana di costante L = 1 rispetto ad x;

  (Infatti: Per ogni x',x" in R: | F(t,x') - F(t,x") | = | x' - x" |.)

- La funzione F(t,0) = 0 è limitata su R;

ma la condizione (b) *non lo è*:

- G(t,x) = x non è limitata su R x R.

Il Teorema di convergenza (30.01) *non è utilizzabile*.

(32.03) Teorema (convergenza del metodo di Eulero esplicito, 2).

Siano t0 un numero reale, F una funzione definita in R x R(n) a valori in R(n) e x0 in R(n) e si consideri il Problema di Cauchy:

(§) x'(t) = F(t,x(t)) , x(t0) = x0

Se tutte le derivate parziali prime di F(t,x) sono funzioni continue di t ed x e il Problema (§) ha una sola soluzione, *allora* per ogni tf > t0 e λ > 0 il metodo di Eulero esplicito applicato al Problema (§) è *convergente per* E → 0.

(Dimostrazione: Omessa.)

(32.04) Realizzazione in Scilab (TS_1_pv).

```
function [T, X, PASSO] = TS_1_pv(x0, t0, tf, F, G2, E, LAMBDA, HMIN)
//
// Integra numericamente, sull'intervallo [t0,tf], il Problema
// di Cauchy in R(n):
//
// x' = F(t,x)
// x(t0) = x0
//
// con il metodo TS(1) - Eulero esplicito - a passo variabile.
//
// x0: condizione iniziale (colonna di n elementi)
// t0: istante iniziale (numero reale)
// tf: istante finale (numero reale)
// F: function che definisce l'equazione differenziale; F(t,x) deve
//    essere una colonna di n numeri reali
// G2: function che restituisce la derivata seconda in t della soluzione
```

```
//     dell'equazione differenziale che all'istante t assume valore x;
//     G2(t,x) deve essere una colonna di n numeri reali
// E: valore massimo della stima dell'errore locale (numero reale)
// LAMBDA: numero reale che stabilisce il valore massimo del passo
//         (OPZIONALE - valore predefinito: 1d-5)
// HMIN: valore minimo consentito del passo
//       (OPZIONALE - valore predefinito: (tf - t0) / 1d6)
//
// T = [t(0),...,t(N)]: riga contenente gli istanti di integrazione
// X = [x(0),...,x(N)]: matrice n x (N+1) contenente le approssimazioni
// PASSO = [h(0),...,h(N-1)]: riga contenente i passi di integrazione
//
// Valore degli argomenti opzionali
//
if ~exists('LAMBDA','l') then LAMBDA = 1d-5; end;
if ~exists('HMIN','l') then HMIN = (tf - t0) / 1d6; end;
//
// Inizializzazione delle variabili di uscita
//
T(1,1) = t0;
X(:,1) = x0;
PASSO = [];
//
// ciclo principale
//
while (T(1,$) < tf), // arresta la costruzione se ha raggiunto tf
  //
  // scelta del passo
  //
  Nd2x = norm(G2(T(1,$),X(:,$)));
  d = max(LAMBDA, Nd2x);
  PASSO(1,$+1) = min(sqrt(2*E/d), tf - T(1,$));
  //
  // calcolo approssimazione e nuovo istante di integrazione
  //
  X(:,$+1) = X(:,$) + F(T(1,$),X(:,$)) * PASSO(1,$);
  T(1,$+1) = T(1,$) + PASSO(1,$);
  //
  // arresta la costruzione se il passo calcolato risulta troppo
  // piccolo e non ha raggiunto tf
  //
  if (PASSO(1,$) < HMIN) & (T(1,$) < tf) then break; end;
  //
end;
//
// Verifica se l'integrazione ha raggiunto tf
//
if T(1,$) < tf then
   printf("\n\nIntegrazione interrotta a T = %3.2e", T(1,$));
end;
//
endfunction
```

```
//
// Nota sui valori di LAMBDA e HMIN
//
//    [T,X,PASSO] = TS_1_pv(x0,t0,tf,F,G,E);
//
//         => LAMBDA = valore predefinito, HMIN = valore predefinito
//
//    [T,X,PASSO] = TS_1_pv(x0,t0,tf,F,G,E,y)
//
//    equivalente a
//
//    [T,X,PASSO] = TS_1_pv(x0,t0,tf,F,G,E,LAMBDA = y)
//
//         => LAMBDA = y, HMIN = valore predefinito
//
//    [T,X,PASSO] = TS_1_pv(x0,t0,tf,F,G,E,y,z)
//
//    equivalente a
//
//    [T,X,PASSO] = TS_1_pv(x0,t0,tf,F,G,E,LAMBDA = y,HMIN = z)
//
//         => LAMBDA = y, HMIN = z
//
//    [T,X,PASSO] = TS_1_pv(x0,t0,tf,F,G,E,HMIN = z)
//
//         => LAMBDA = valore predefinito, HMIN = z
//
```